
TP Visual Basic n°3 & 4


Utilisation des DAO

Au cours de cet atelier, vous allez créer et tester des feuilles qui utilisent des objets d'accès aux données (DAO).

À l'issue de cet atelier, vous serez en mesure d'effectuer les tâches suivantes : 


ouvrir une base de données à l'aide des objets DAO ;


naviguer dans un jeu d'enregistrements et le manipuler à l'aide des objets DAO ;


retrouver des informations dans une table ;


créer un jeu d'enregistrements à partir d'une requête ;


utiliser une requête avec paramètres.

TP n°3 : manipulation d'un jeu d'enregistrements

Dans cette phase, vous allez utiliser des objets d'accès aux données (DAO) pour ouvrir une base de données et créer un jeu d'enregistrements.



Créez une feuille Catégories


1.
Créez un nouveau projet EXE standard dans Visual Basic.


2.
Choisissez le nom frmCategories pour la feuille par défaut.


3.
Ajoutez des contrôles à la feuille, comme indiqué sur l'illustration qui suit.

Remarque  Étant donné que les informations d'identification ne sont pas modifiées par l'utilisateur, le champ Code catégorie est un contrôle Label.



Ouvrez une base de données et créez un jeu d'enregistrements

1.  Dans la feuille frmCategories, déclarez deux variables de niveau module : dbCurrent ( pour récupérer le nom de la base de données ) et , recCategories ( le jeu d’enregistrement ).

2.  Définissez une référence à la bibliothèque d'objets Microsoft DAO 3.5.


3.
Dans la procédure événementielle Load de la feuille Catégories vous devez :


a.
Ouvrir le fichier Comptoir.mdb et enregistrer l'objet de base de données résultant dans la variable dbCurrent.


b.
Créer un jeu d'enregistrements à l'aide de la table Catégories de la base de données Comptoir.mdb et stocker l'objet résultant dans la variable recCategories.

c.  Se positionner sur le premier enregistrement du jeu d'enregistrements ( utiliser la méthode Fields des objets RecordSet pour l’affichage )

d.  Activater des boutons de commande Ajouter, Modifier, Supprimer, Rechercher, Précédent, et Suivant.

e.  Désactiver les zones de texte.

f.  Désactiver les boutons Enregistrer et Annuler.

4.  Dans la procédure événementielle Unload, fermez la base de données. En déduire le code du bouton Fermer.

5.  Testez l'application.



Écrivez le code correspondant aux différents boutons.

1.  Ajoutez le code voulu pour activer les boutons de commande Précédent et Suivant ( pensez au début et la fin du jeu d'enregistrements). Pendant qu'un utilisateur navigue dans la feuille Catégories, désactivez les zones de texte, ainsi que les boutons Enregistrer et Annuler, afin que des modifications ne puissent être apportées aux données que si l'utilisateur sélectionne le bouton Modifier ou Ajouter.

2.   Pour effacer les champs de texte et passer en mode Ajout, écrivez le code voulu pour le bouton Ajouter.Lorsqu'un utilisateur sélectionne le bouton Ajouter ou Modifier, activez les zones de texte, ainsi que les boutons Enregistrer et Annuler. Désactivez tous les autres boutons sur la feuille.

3.  Pour passer en mode Édition, écrivez le code voulu pour le bouton Modifier. 

4.  Pour répercuter les modifications apportées aux zones de texte dans le jeu d'enregistrements et mettre ce dernier à jour, écrivez le code correspondant au bouton Enregistrer.

5.  Pour annuler la mise à jour en attente et mettre les champs à jour à partir de l'enregistrement en cours dans le jeu d'enregistrements, écrivez le code correspondant au bouton Annuler.

6.  Pour supprimer l'enregistrement en cours, écrivez le code correspondant au bouton Supprimer. Placez-vous sur l'enregistrement suivant et procédez à l'extraction des champs. Gérez les erreurs dûes aux violations d’intégrité référentielle.

7.  Testez l’application



Écrivez le code correspondant au bouton Rechercher


1.
Utilisez la fonction InputBox pour inviter les utilisateurs à spécifier une chaîne qui représente une partie du champ Description.

2.  À partir de la chaîne spécifiée, créez une chaîne SQL afin de sélectionner les enregistrements correspondants dans la table Catégories. Créez un jeu d'enregistrements basé sur la requête SQL, et affectez les résultats à la variable recCategories. Ceci active les autres boutons dans le nouveau jeu d'enregistrements. Si aucun enregistrement n'est trouvé (recCategories.RecordCount = 0), affichez un message pour l'indiquer. Dans le cas contraire, affichez le premier enregistrement dans le nouveau jeu d'enregistrements.

3.  Testez l’application : tapez fruits à l'invite ( deux enregistrements dont la description contient le texte fruits doivent normalement s'afficher ).

TP n°4 : Utilisation d’UNE requête

Dans cette phase, vous allez renvoyer des informations contenues dans la base de données Comptoir.mdb à l'aide d'une requête existante.



 Demandez les dix produits les plus chers


1.
Créez une nouvelle feuille.

2.  La requête Les dix produits les plus chers présente dans la base de données Comptoir.mdb renvoie les dix produits les plus chers. Dans la procédure événementielle Form_Load de la nouvelle feuille, créez un jeu d'enregistrements basé sur cette requête, et affichez les résultats dans un contrôle DBGrid.
3.  Testez l’application.

SOLUTIONS

‘

‘ Général

‘

Dim DbCurrent As Database

Dim RecCategories As Recordset

Private Sub Init_Ajout_Modif()

  NomCateg.Enabled = True

  NomCateg.SetFocus

  DesCateg.Enabled = True

  BtnModifier.Enabled = False

  BtnAjouter.Enabled = False

  BtnEnregistrer.Enabled = True

  BtnAnnuler.Enabled = True

  BtnRechercher.Enabled = False

  BtnSupprimer.Enabled = False

End Sub

Private Sub Initialisation()

'

' Les boutons

'

   BtnModifier.Enabled = True

   BtnAjouter.Enabled = True

   BtnEnregistrer.Enabled = False

   BtnAnnuler.Enabled = False

   BtnSupprimer.Enabled = True

   BtnRechercher.Enabled = True

'

' Les zones de texte

'

   NomCateg.Enabled = False

   DesCateg.Enabled = False

End Sub

Private Sub RemplitChamps()

   CodeCateg.Caption = RecCategories.Fields("Code Catégorie")

   NomCateg.Text = RecCategories.Fields("Nom de catégorie")

   DesCateg.Text = RecCategories.Fields("Description")

End Sub

‘

‘ Procédures liées aux objets

‘

Private Sub Form_Load()

   Set DbCurrent = OpenDatabase("D:\mvb5\labs\comptoir.mdb")

   Set RecCategories = DbCurrent.OpenRecordset("Catégories")

   RecCategories.MoveFirst

   RemplitChamps

   Initialisation

End Sub

Private Sub BtnPrecedent_Click()

  RecCategories.MovePrevious

  If RecCategories.BOF Then

    Beep

    MsgBox ("Premier enregistrement ! ")

    RecCategories.MoveFirst

  Else

    RemplitChamps

  End If

  Initialisation

End Sub

Private Sub BtnSuivant_Click()

  RecCategories.MoveNext

  If RecCategories.EOF Then

    Beep

    MsgBox ("Dernier enregistrement ! ")

    RecCategories.MovePrevious

  Else

    RemplitChamps

  End If

  Initialisation

End Sub

Private Sub BtnAjouter_Click()

  Set RecCategories = DbCurrent.OpenRecordset("Catégories")

' Le Set précédent permet de s'assurer que l'on travaille sur la table entière

' et non sur une partie de la table ( après une recherche fructueuse en particulier )

  RecCategories.AddNew

  CodeCateg.Caption = RecCategories.Fields("Code Catégorie")

  NomCateg.Text = ""

  DesCateg.Text = ""

  Init_Ajout_Modif

End Sub

	Private Sub BtnModifier_Click()

  Init_Ajout_Modif

  RecCategories.Edit

End Sub
	Private Sub BtnAnnuler_Click()
   RecCategories.CancelUpdate

   RemplitChamps

   Initialisation

End Sub
	Private Sub BtnFermer_Click()
  DbCurrent.Close

  Unload Me

End Sub


Private Sub BtnEnregistrer_Click()

  RecCategories.Fields("Nom de catégorie") = NomCateg.Text

  RecCategories.Fields("Description") = DesCateg.Text

  RecCategories.Update

  RecCategories.Bookmark = RecCategories.LastModified

  Initialisation

End Sub

Private Sub BtnRechercher_Click()

Dim Chaine, ReqSql As String

  Chaine = InputBox("Texte à rechercher dans la description : ")

  ReqSql = "Select * From Catégories Where Description like " & "'*" & Chaine & "*'"

  Set RecCategories = DbCurrent.OpenRecordset(ReqSql)

  If RecCategories.RecordCount = 0 Then

    MsgBox ("Aucun enregistrement trouvé...")

    Set RecCategories = DbCurrent.OpenRecordset("Catégories")

  Else

    RecCategories.MoveFirst

    RemplitChamps

  End If

End Sub

Private Sub BtnSupprimer_Click()

  On Error GoTo TraitErreur

  RecCategories.Delete

  RecCategories.MoveNext

  If RecCategories.EOF Then

    RecCategories.MovePrevious

  End If

  RemplitChamps

  Exit Sub

TraitErreur:

  If Err.Number = 3200 Then

    MsgBox ("Violation des règles d'intégrités référentielles")

  Else

    MsgBox ("Ca plante grave ...")

  End If

End Sub

